Combining predicate transformer semantics for effects: a case study in parsing regular languages

Tim Baanen
Vrije Universiteit Amsterdam

Wouter Swierstra
Utrecht University

This paper describes how to verify a parser for regular expressions in a functional programming language using predicate transformer semantics for a variety of effects. Where our previous work in this area focused on the semantics for a single effect, parsing requires a combination of effects: non-determinism, general recursion and mutable state. Reasoning about such combinations of effects is notoriously difficult, yet our approach using predicate transformers enables the careful separation of program syntax, correctness proofs and termination proofs.

1 Introduction

There is a significant body of work on parsing using combinators in functional programming languages [30, 12, 8, 27, 16, 14, 9, 19, among many others]. Yet how can we ensure that these parsers are correct? There is notably less work that attempts to answer this question [5, 7].

Reasoning about such parser combinators is not at all trivial. They use a variety of effects: state to store the string being parsed; non-determinism to handle backtracking; and general recursion to deal with recursive grammars. Proof techniques, such as equational reasoning, that are commonly used when reasoning about pure functional programs, are less suitable when verifying effectful programs [10, 13].

In this paper, we explore a novel approach, drawing inspiration from recent work on algebraic effects [3, 31, 18]. We demonstrate how to reason about all parsers uniformly using predicate transformers [29]. We extend our previous work that uses predicate transformer semantics to reason about a single effect, to handle the combinations of effects used by parsers. Our semantics is modular, meaning we can introduce new effects, semantics and specifications when they are needed, without having to rework the previous definitions. In particular, our careful treatment of general recursion lets us separate partial correctness from the proof of termination cleanly. Most existing proofs require combinators to guarantee that the string being parsed decreases, conflating these two issues.

In particular, the sections of this paper make the following contributions:

- After quickly revisiting our previous work on predicate transformer semantics for effects (Section 2), we show how the non-recursive fragment of regular expressions can be correctly parsed using non-determinism (Section 3).
- By combining non-determinism with general recursion (Section 4), support for the Kleene star can be added without compromising our previous definitions (Section 5).
- Although the resulting parser is not guaranteed to terminate, we can define another implementation using Brzozowski derivatives (Section 6), introducing an additional effect and its semantics in the process.

Submitted to:
MSFP 2020

- Finally, we show that the derivative-based implementation terminates and refines the original parser (Section 7).

The goal of our work is not so much the verification of a parser for regular languages, which has been done before [11, 15]. Instead, we aim to illustrate the steps of incrementally developing and verifying a parser using predicate transformers and algebraic effects. This work is in the spirit of a Theoretical Pearl [11]: we begin by defining a *match* function that does not terminate. The remainder of the paper then shows how to fix this function, without having to redo the complete proof of correctness.

All the programs and proofs in this paper are written in the dependently typed language Agda [21]. The full source code, including lemmas we have chosen to omit for sake of readability, is available online.[1] Apart from postulating function extensionality, we remain entirely within Agda's default theory.

2 Recap: algebraic effects and predicate transformers

Algebraic effects separate the syntax and semantics of effectful operations. In this paper, we will model them by taking the free monad over a given signature, describing certain operations. The type of such a signature is defined as follows:

```
record Sig : Set where
  constructor mkSig
  field
    C : Set
    R : C → Set
```

Here the type C contains the 'commands', or effectful operations that a given effect supports. For each command $c : C$, the type $R\,c$ describes the possible responses. The structure on a signature is that of a container [1]. The following signature describes two operations: the non-deterministic choice between two values, *Choice*; and a failure operator, *Fail*.

```
data CNondet : Set where
  Choice : CNondet
  Fail   : CNondet


RNondet : CNondet → Set
RNondet Choice = Bool
RNondet Fail   = ⊥
Nondet = mkSig CNondet RNondet
```

We represent effectful programs that use a particular effect using the corresponding free monad:

```
data Free (e : Sig) (a : Set) : Set where
  Pure : a → Free e a
  Op : (c : C e) → (R e c → Free e a) → Free e a
```

[1] https://github.com/Vierkantor/refinement-parsers

This gives a monad, with the bind operator defined as follows.

$$
\begin{aligned}
&\_ \gg\!= \_ : \mathit{Free}\ e\ a \to (a \to \mathit{Free}\ e\ b) \to \mathit{Free}\ e\ b \\
&\mathit{Pure}\ x \gg\!= f = f\ x \\
&\mathit{Op}\ c\ k \gg\!= f = \mathit{Op}\ c\ (\lambda\ x \to k\ x \gg\!= f)
\end{aligned}
$$

To facilitate programming with effects, we define the following smart constructors, sometimes referred to as generic effects in the literature [22]:

$$
\begin{aligned}
&\mathit{fail} : \mathit{Free}\ \mathit{Nondet}\ a \\
&\mathit{fail} = \mathit{Op}\ \mathit{Fail}\ \lambda\ () \\
&\mathit{choice} : \mathit{Free}\ \mathit{Nondet}\ a \to \mathit{Free}\ \mathit{Nondet}\ a \to \mathit{Free}\ \mathit{Nondet}\ a \\
&\mathit{choice}\ S_1\ S_2 = \mathit{Op}\ \mathit{Choice}\ \lambda\ b \to \text{if } b \text{ then } S_1 \text{ else } S_2
\end{aligned}
$$

In this paper, we will assign semantics to effectful programs by mapping them to predicate transformers. Each semantics will be computed by a fold over the free monad, mapping some predicate $P : a \to \mathit{Set}$ to a predicate on the result of the free monad to a predicate of the entire computation of type $\mathit{Free}\ (\mathit{eff}\ C\ R)\ a \to \mathit{Set}$.

$$
\begin{aligned}
&[\![\_]\!] : ((c : C) \to (R\ c \to \mathit{Set}) \to \mathit{Set}) \to \mathit{Free}\ (\mathit{mkSig}\ C\ R)\ a \to (a \to \mathit{Set}) \to \mathit{Set} \\
&[\![\mathit{Pure}\ x]\!]_{\mathit{alg}}\ P = P\ x \\
&[\![\mathit{Op}\ c\ k]\!]_{\mathit{alg}}\ P = \mathit{alg}\ c\ \lambda\ x \to [\![k\ x]\!]_{\mathit{alg}}\ P
\end{aligned}
$$

The predicate transformer nature of these semantics becomes evident when we assume the type of responses R does not depend on the command $c : C$. The type of $\mathit{alg} : (c : C) \to (R\ c \to \mathit{Set}) \to \mathit{Set}$ then becomes $C \to (R \to \mathit{Set}) \to \mathit{Set}$, which is isomorphic to $(R \to \mathit{Set}) \to (C \to \mathit{Set})$. Thus, alg has the form of a predicate transformer from postconditions of type $R \to \mathit{Set}$ into preconditions of type $C \to \mathit{Set}$.

Two considerations cause us to define the types $\mathit{alg} : (c : C) \to (R\ c \to \mathit{Set}) \to \mathit{Set}$, and analogously $[\![\_]\!]_{\mathit{alg}} : \mathit{Free}\ (\mathit{mkSig}\ C\ R)\ a \to (a \to \mathit{Set}) \to \mathit{Set}$. By having the command as first argument to alg, we allow R to depend on C. Moreover, $[\![\_]\!]_{\mathit{alg}}$ computes semantics, so it should take a program $S : \mathit{Free}\ (\mathit{mkSig}\ C\ R)\ a$ as its argument and return the semantics of S, which is then of type $(a \to \mathit{Set}) \to \mathit{Set}$.

In the case of non-determinism, for example, we may want to require that a given predicate P holds for all possible results that may be returned:

$$
\begin{aligned}
&\mathit{ptAll} : (c : \mathit{CNondet}) \to (\mathit{RNondet}\ c \to \mathit{Set}) \to \mathit{Set} \\
&\mathit{ptAll}\ \mathit{Fail}\ P \quad = \top \\
&\mathit{ptAll}\ \mathit{Choice}\ P = P\ \mathit{True} \wedge P\ \mathit{False}
\end{aligned}
$$

A different semantics may instead require that P holds on any of the return values:

$$
\begin{aligned}
&\mathit{ptAny} : (c : \mathit{CNondet}) \to (\mathit{RNondet}\ c \to \mathit{Set}) \to \mathit{Set} \\
&\mathit{ptAny}\ \mathit{Fail}\ P \quad = \bot \\
&\mathit{ptAny}\ \mathit{Choice}\ P = P\ \mathit{True} \vee P\ \mathit{False}
\end{aligned}
$$

Predicate transformers provide a single semantic domain to relate programs and specifications [20]. Throughout this paper, we will consider specifications consisting of a pre- and postcondition:

```
record Spec (a : Set) : Set where
  constructor [_,_]
  field
    pre : Set
    post : a → Set
```

Inspired by work on the refinement calculus, we can assign a predicate transformer semantics to specifications as follows:

```
⟦_,_⟧spec : Spec a → (a → Set) → Set
⟦pre,post⟧spec P = pre ∧ (∀ o → post o → P o)
```

This computes the 'weakest precondition' necessary for a specification to imply that the desired postcondition *P* holds. In particular, the precondition *pre* should hold and any possible result satisfying the postcondition *post* should imply the postcondition *P*.

Finally, we use the refinement relation to compare programs and specifications:

```
_⊑_ : (pt1 pt2 : (a → Set) → Set) → Set
pt1 ⊑ pt2 = ∀ P → pt1 P → pt2 P
```

Together with the predicate transformer semantics we have defined above, this refinement relation can be used to relate programs to their specifications. The refinement relation is both transitive and reflexive.

3 Regular languages without recursion

To illustrate how to reason about non-deterministic code, we begin by defining a regular expression matcher. Initially, we will restrict ourselves to non-recursive regular expressions; we will add recursion in the next section.

We begin by defining the *Regex* datatype for regular expressions as follows: An element of this type represents the syntax of a regular expression.

```
data Regex : Set where
  Empty     : Regex
  Epsilon   : Regex
  Singleton : Char → Regex
  _|_       : Regex → Regex → Regex
  _·_       : Regex → Regex → Regex
  _⋆        : Regex → Regex
```

Note that the *Empty* regular expression corresponds to the empty language, while the *Epsilon* expression only matches the empty string. Furthermore, our language for regular expressions is closed under choice (_|_), concatenation (_·_) and linear repetition denoted by the Kleene star (_⋆).

What should our regular expression matcher return? A Boolean value is not particularly informative; yet we also choose not to provide an intrinsically correct definition, instead performing extrinsic verification using our predicate transformer semantics. The Tree data type below, captures a potential parse tree associated with a given regular expression:

```
Tree : Regex → Set
Tree Empty          = ⊥
Tree Epsilon        = ⊤
Tree (Singleton _)  = Char
Tree (l | r)        = Either (Tree l) (Tree r)
Tree (l · r)        = Pair (Tree l) (Tree r)
Tree (r ⋆)          = List (Tree r)
```

In the remainder of this section, we will develop a regular expression matcher with the following type:

```
match : (r : Regex) (xs : String) → Free Nondet (Tree r)
```

Before we do so, however, we will complete our specification. Although the type above guarantees that we return a parse tree matching the regular expression *r*, there is no relation between the tree and the input string. To capture this relation, we define the following *Match* data type. A value of type *Match r xs t* states that the string *xs* is in the language given by the regular expression *r* as witnessed by the parse tree *t*:

```
data Match : (r : Regex) → String → Tree r → Set where
  Epsilon     : Match Epsilon Nil tt
  Singleton   : Match (Singleton x) (x :: Nil) x
  OrLeft      : Match l xs x → Match (l | r) xs (Inl x)
  OrRight     : Match r xs x → Match (l | r) xs (Inr x)
  Concat      : Match l ys y → Match r zs z → Match (l · r) (ys ++ zs) (y , z)
  StarNil     : Match (r ⋆) Nil Nil
  StarConcat : Match (r · (r ⋆)) xs (y , ys) → Match (r ⋆) xs (y :: ys)
```

Note that there is no constructor for *Match Empty xs ms* for any *xs* or *ms*, as there is no way to match the *Empty* language with a string *xs*. Similarly, the only constructor for *Match Epsilon xs ms* is where *xs* is the empty string *Nil*. There are two constructors that produce a *Match* for a regular expression of the form $l \mid r$, corresponding to the choice of matching either l or r.

The cases for concatenation and iteration are more interesting. Crucially the *Concat* constructor constructs a match on the concatenation of the strings *xs* and *zs* – although there may be many possible ways to decompose a string into two substrings. Finally, the two constructors for the Kleene star, $r \star$, match zero (*StarNil*) or many (*StarConcat*) repetitions of *r*.

We will now turn our attention to the *match* function. The complete definition, by induction on the argument regular expression, can be found in Figure 1. Most of the cases are straightforward—the most difficult case is that for concatenation, where we non-deterministically consider all possible splittings of the input string *xs* into a pair of strings *ys* and *zs*. The *allSplits* function, defined below, computes all possible splittings:

```
allSplits : (xs : List a) → Free Nondet (List a × List a)
allSplits Nil = Pure (Nil , Nil)
allSplits (x :: xs) = choice
  (Pure (Nil , (x :: xs)))
  (allSplits xs >>= λ {(ys , zs) → Pure ((x :: ys) , zs)})
```

```
match : (r : Regex) (xs : String) → Free Nondet (Tree r)
match Empty       xs          = fail
match Epsilon     Nil         = Pure tt
match Epsilon     (_ :: _)    = fail
match (Singleton c) Nil       = fail
match (Singleton c) (x :: Nil)  with c ≟ x
match (Singleton c) (.c :: Nil) | yes refl = Pure c
match (Singleton c) (x :: Nil)  | no ¬p    = fail
match (Singleton c) (_ :: _ :: _) = fail
match (l ∣ r)     xs          = choice (Inl ⟨$⟩ match l xs) (Inr ⟨$⟩ match r xs)
match (l · r)     xs          = do (ys , zs) ← allSplits xs
                                   y ← match l ys
                                   z ← match r zs
                                   Pure (y , z)
match (r ⋆) xs                = fail
```

Figure 1: The definition of the *match* function

Finally, we cannot yet implement the case for the Kleene star. We could attempt to mimic the case for concatenation, attempting to match $r \cdot (r\star)$. This definition, however, is rejected by Agda as it is not structurally recursive. For now we choose to simply fail on all such regular expressions.

Still, we can prove that the *match* function behaves correctly on all regular expressions that do not contain iteration. We introduce a *hasNo*$*$ predicate, which holds of all such iteration-free regular expressions:

```
hasNo* : Regex → Set
```

To verify our matcher is correct, we need to prove that it satisfies the specification consisting of the following pre- and postcondition:

```
pre : (r : Regex) (xs : String) → Set
pre r xs = hasNo* r
post : (r : Regex) (xs : String) → Tree r → Set
post = Match
```

The main correctness result can now be formulated as follows:

$$\mathit{matchSound} : \forall\ r\ xs \rightarrow [\![(\mathit{pre}\ r\ xs), (\mathit{post}\ r\ xs)]\!]_{\mathrm{spec}} \sqsubseteq [\![\mathit{match}\ r\ xs]\!]_{\mathit{ptAll}}$$

This lemma guarantees that all the parse trees computed by the *match* function satisfy the *Match* relation, provided the input regular expression does not contain iteration. The proof goes by induction on the regular expression r. Although we have omitted the proof, we will sketch the key lemmas and definitions that are necessary to complete it.

In most of the cases for r, the definition of *match r* is uncomplicated and the proof is similarly simple. As soon as we need to reason about programs composed using the monadic bind operator,

we quickly run into issues. In particular, when verifying the case for $l \cdot r$, we would like to use our induction hypotheses on two recursive calls. To do, we prove the following lemma that allows us to replace the semantics of a composite program built using the monadic bind operation with the composition of the underlying predicate transformers:

$$\begin{aligned}&\mathit{consequence} : \forall\, \mathit{pt}\ (\mathit{mx} : \mathit{Free}\ \mathit{es}\ a)\ (f : a \to \mathit{Free}\ \mathit{es}\ b) \to \\ &\quad [\![\mathit{mx}]\!]_{\mathit{pt}}\ (\lambda\, x \to [\![f\ x]\!]_{\mathit{pt}}\ P) \equiv [\![\mathit{mx} \gg\!= f]\!]_{\mathit{pt}}\ P\end{aligned}$$

Substituting along this equality gives us the lemmas we need to deal with the _>>=_ operator:

$$\begin{aligned}&\mathit{wpToBind} : (\mathit{mx} : \mathit{Free}\ \mathit{es}\ a)\ (f : a \to \mathit{Free}\ \mathit{es}\ b) \to \\ &\quad [\![\mathit{mx}]\!]_{\mathit{pt}}\ (\lambda\, x \to [\![f\ x]\!]_{\mathit{pt}}\ P) \to [\![\mathit{mx} \gg\!= f]\!]_{\mathit{pt}}\ P \\ &\mathit{wpFromBind} : (\mathit{mx} : \mathit{Free}\ \mathit{es}\ a)\ (f : a \to \mathit{Free}\ \mathit{es}\ b) \to \\ &\quad [\![\mathit{mx} \gg\!= f]\!]_{\mathit{pt}}\ P \to [\![\mathit{mx}]\!]_{\mathit{pt}}\ (\lambda\, x \to [\![f\ x]\!]_{\mathit{pt}}\ P)\end{aligned}$$

Not only does *match* $(l \cdot r)$ result in two recursive calls, it also makes a call to a helper function *allSplits*. Thus, we also need to formulate and prove the correctness of that function, as follows:

$$\begin{aligned}&\mathit{allSplitsPost} : \mathit{String} \to \mathit{String} \times \mathit{String} \to \mathit{Set} \\ &\mathit{allSplitsPost}\ \mathit{xs}\ (\mathit{ys}\ ,\ \mathit{zs}) = \mathit{xs} \equiv \mathit{ys} +\!\!+ \mathit{zs} \\ &\mathit{allSplitsSound} : \forall\, \mathit{xs} \to [\![\top, (\mathit{allSplitsPost}\ \mathit{xs})]\!]_{\mathrm{spec}} \sqsubseteq [\![\mathit{allSplits}\ \mathit{xs}]\!]_{\mathit{ptAll}}\end{aligned}$$

Using *wpToBind*, we can incorporate the correctness proof of *allSplits* in the correctness proof of *match*. We refer to the accompanying code for the complete details of these proofs.

4 General recursion and non-determinism

The matcher we have defined in the previous section is incomplete, since it fails to handle regular expressions that use the Kleene star. The fundamental issue is that the Kleene star allows for arbitrarily many matches in certain cases, that in turn, leads to problems with Agda's termination checker. For example, matching $\mathit{Epsilon}\ \star$ with the empty string `""` may unfold the Kleene star infinitely often without ever terminating. As a result, we cannot implement *match* for the Kleene star using recursion directly.

Instead, we will deal with this (possibly unbounded) recursion by introducing a new effect. We will represent a recursively defined (dependent) function of type $(i : I) \to O\ i$ as an element of the type $(i : I) \to \mathit{Free}\ (\mathit{Rec}\ I\ O)\ (O\ i)$. Here $\mathit{Rec}\ I\ O$ is a synonym of the the signature type we saw previously [18]:

$$\begin{aligned}&\mathit{Rec} : (I : \mathit{Set})\ (O : I \to \mathit{Set}) \to \mathit{Sig} \\ &\mathit{Rec}\ I\ O = \mathit{mkSig}\ I\ O\end{aligned}$$

Intuitively, you may want to think of values of type $(i : I) \to \mathit{Free}\ (\mathit{Rec}\ I\ O)\ (O\ i)$ as computing a (finite) call graph for every input $i : I$. Instead of recursing directly, the 'effects' that this signature supports require an input $i : I$ corresponding to the argument of the recursive call; the continuation abstracts over a value of type $O\ i$, corresponding to the result of a recursive call. Note that the functions defined in this style are not recursive; instead we will need to write handlers to unfold the function definition or prove termination separately.

We cannot, however, define a *match* function of the form *Free* (*Rec* _ _) directly, as our previous definition also used non-determinism. To account for both non-determinism and unbounded recursion, we need a way to combine effects. Fortunately, free monads are known to be closed under coproducts; there is a substantial body of work that exploits this to (syntactically) compose separate effects [31, 28].

Rather than restrict ourselves to the binary composition using coproducts, we modify the *Free* monad to take a list of signatures as its argument, taking the coproduct of the elements of the list as its signature functor. The *Pure* constructor remains as unchanged; while the *Op* constructor additionally takes an index into the list to specify the effect that is invoked.

```
data Free (es : List Sig) (a : Set) : Set where
  Pure : a → Free es a
  Op : (i : e ∈ es) (c : C e) (k : R e c → Free es a) → Free es a
```

By using a list of effects instead of allowing arbitrary disjoint unions, we have effectively chosen that the disjoint unions canonically associate to the right. We choose to use the same names and (almost) the same syntax for this new definition of *Free*, since all the definitions that we have seen previously can be readily adapted to work with this data type instead.

Most of this bookkeeping involved with different effects can be inferred using Agda's instance arguments [6]. Instance arguments, marked using the double curly braces {{ }}, are automatically filled in by Agda, provided a unique value of the required type can be found. For example, we can define the generic effects that we saw previously as follows:

```
fail : {{ iND : Nondet ∈ es }} → Free es a
fail {{ iND }} = Op iND Fail λ ()
choice : {{ iND : Nondet ∈ es }} → Free es a → Free es a → Free es a
choice {{ iND }} S1 S2 = Op iND Choice λ b → if b then S1 else S2
call : {{ iRec : Rec I O ∈ es }} → (i : I) → Free es (O i)
call {{ iRec }} i = Op iRec i Pure
```

These now operate over any free monad with effects given by *es*, provided we can show that the list *es* contains the *Nondet* and *Rec* effects respectively. For convenience of notation, we introduce the $\_ \overset{es}{\nrightarrow} \_$ notation for the type of generally recursive functions with effects in *es*, i.e. Kleisli arrows into *Free* (*Rec* _ _ :: *es*).

```
_ ↛ _ : (I : Set) (es : List Sig) (O : I → Set) → Set
I ↛[es] O = (i : I) → Free (Rec I O :: es) (O i)
```

With the syntax for combinations of effects defined, let us turn to semantics. Since the weakest precondition predicate transformer for a single effect is given as a fold over the effect's signature, the semantics for a combination of effects can be given by a list of such semantics.

```
record PT (e : Sig) : Set where
  constructor mkPT
  field
    pt   : (c : C e) → (R e c → Set) → Set
    mono : ∀ c P P′ → P ⊆ P′ → pt c P → pt c P′
```

```
data PTs : List Sig → Set where
  Nil    : PTs Nil
  _::_ : PT e → PTs es → PTs (e :: es)
```

The record type *PT* not only contains a predicate transformer *pt*, but also a proof that this predicate transformer is monotone. Several lemmas throughout this paper, such as the terminates-fmap lemma below, rely on the monotonicity of the underlying predicate transformers; for each semantics we present the proof of monotonicity is immediate.

Given a such a list of predicate transformers, defining the semantics of an effectful program is a straightforward generalization of the previously defined semantics. The *Pure* case is identical, and in the *Op* case we can apply the predicate transformer returned by the *lookupPT* helper function.

```
lookupPT : (pts : PTs es) (i : mkSig C R ∈ es) → (c : C) → (R c → Set) → Set
lookupPT (pt :: pts) (∈Head)   = PT.pt pt
lookupPT (pt :: pts) (∈Tail i) = lookupPT pts i
```

This results in the following definition of the semantics for combinations of effects.

```
⟦_⟧ : (pts : PTs es) → Free es a → (a → Set) → Set
⟦Pure x⟧pts  P = P x
⟦Op i c k⟧pts P = lookupPT pts i c λ x → ⟦k x⟧pts P
```

The effects that we will use for our *match* function consist of a combination of nondeterminism and general recursion. Although we can reuse the *ptAll* semantics of nondeterminism, we have not yet given the semantics for recursion. However, it is not as easy to give a predicate transformer semantics for recursion in general, since the intended semantics of a recursive call depend on the function that is being defined. Instead, to give semantics to a recursive function, we assume that we have been provided with a relation of the type $(i : I) \to O\ i \to Set$, reminiscent of a loop invariant in an imperative program. The semantics then establishes whether or not the recursive function adheres to this invariant or not:

```
ptRec : ((i : I) → O i → Set) → PT (Rec I O)
PT.pt (ptRec R) i P = ∀ o → R i o → P o
```

As we shall see shortly, when revisiting the *match* function, the *Match* relation defined previously will fulfill the role of this 'invariant.'

5 Recursively parsing every regular expression

To deal with the Kleene star, we rewrite *match* as a generally recursive function using a combination of effects. Since *match* makes use of *allSplits*, we also rewrite that function to use a combination of effects. The types become:

```
allSplits : {{ iND : Nondet ∈ es }} → List a → Free es (List a × List a)
match : {{ iND : Nondet ∈ es }} → (x : Regex × String) ↬es Tree (Pair.fst x)
```

Since the index argument to the smart constructor is inferred by Agda, the only change in the definition of *match* and *allSplits* will be that *match* now does have a meaningful branch for the Kleene star case:

```
match ((r ⋆) , Nil) = Pure Nil
match ((r ⋆) , xs@ (_ :: _)) = do
  (y , ys) ← call ((r · (r ⋆)) , xs)
  Pure (y :: ys)
```

The effects we need to use for running *match* are a combination of nondeterminism and general recursion. As discussed, we first need to give the specification for *match* before we can verify a program that performs a recursive *call* to *match*.

```
matchSpec : (r,xs : Pair Regex String) → Tree (Pair.fst r,xs) → Set
matchSpec (r , xs) ms = Match r xs ms

⟦_⟧match : Free (Rec (Pair Regex String) (Tree ∘ Pair.fst) :: Nondet :: Nil) a →
  (a → Set) → Set
⟦S⟧match = ⟦S⟧ptRec matchSpec :: ptAll :: Nil
```

We can reuse exactly our proof that *allSplits* is correct, since we use the same semantics for the non-determinism used in the definition of *allSplits*. Similarly, the partial correctness proof of *match* will be the same on all cases except the Kleene star. Now we are able to prove correctness of *match* on a Kleene star.

```
matchSound ((r ⋆) , Nil)       P (preH , postH)       = postH _ StarNil
matchSound ((r ⋆) , (x :: xs)) P (preH , postH) o H = postH _ (StarConcat H)
```

At this point, we have defined a matcher for regular languages and formally proven that when it succeeds in recognizing a given string, this string is indeed in the language generated by the argument regular expression. However, the *match* function does not necessarily terminate: if r is a regular expression that accepts the empty string, then calling *match* on $r\star$ and a string xs will diverge. In the next section, we will write a new parser that is guaranteed to terminate and show that this parser refines the *match* function defined above.

6 Derivatives and handlers

Since recursion on the structure of a regular expression does not guarantee termination of the parser, we can instead perform recursion on the string to be parsed, changing the regular expression to be matched based on the characters we have seen.

The Brzozowski derivative of a formal language L with respect to a character x consists of all strings xs such that $x :: xs \in L$ [4]. Crucially, if L is regular, so are all its derivatives. Thus, let r be a regular expression, and $d\ r\ /d\ x$ an expression for the derivative with respect to x, then r matches a string $x :: xs$ if and only if $d\ r\ /d\ x$ matches xs. This suggests the following implementation of matching an expression r with a string xs: if xs is empty, check whether r matches the empty string; otherwise remove the head x of the string and try to match $d\ r\ /d\ x$.

The first step in implementing a parser using the Brzozowski derivative is to compute the derivative for a given regular expression. Following Brzozowski [4], we use a helper function $\varepsilon?$ that decides whether an expression matches the empty string.

```
ε? : (r : Regex) → Dec (∑ (Tree r) (Match r Nil))
```

The definition of $\varepsilon?$ is given by structural recursion on the regular expression, just as the derivative operator is:

```
d_/d_ : Regex → Char → Regex
d Empty       /d c = Empty
d Epsilon     /d c = Empty
d Singleton x /d c with c ≟ x
...                 | yes p = Epsilon
...                 | no ¬p = Empty
d l · r       /d c with ε? l
...                 | yes p = ((d l /d c) · r) | (d r /d c)
...                 | no ¬p = (d l /d c) · r
d l | r       /d c = (d l /d c) | (d r /d c)
d r ⋆         /d c = (d r /d c) · (r ⋆)
```

To use the derivative of r to compute a parse tree for r, we need to be able to convert a tree for $d\ r\ /d\ x$ to a tree for r. As this function 'inverts' the result of differentiation, we name it *integralTree*:

```
integralTree : (r : Regex) → Tree (d r /d x) → Tree r
```

Its definition closely follows the pattern matching performed in the definition of $d\_/d\_$.

The description of a derivative-based matcher is stateful: we perform a step by removing a character from the input string. This state can be encapsulated in a new effect *Parser*. The *Parser* effect has one command *Symbol* that returns a *Maybe Char*. Calling *Symbol* will return *just* the head of the unparsed remainder (advancing the string by one character) or *nothing* if the string has been totally consumed.

```
data CParser : Set where
  Symbol : CParser
RParser : CParser → Set
RParser Symbol = Maybe Char
Parser = mkSig CParser RParser

symbol : {{ iP : Parser ∈ es }} → Free es (Maybe Char)
symbol {{ iP }} = Op iP Symbol Pure
```

The code for the parser, *dmatch*, is now only a few lines long. When the input contains at least one character, we use the derivative to match the first character and recurse; when the input string is empty, we check that the expression matches the empty string.

```
dmatch : {{ iP : Parser ∈ es }} {{ iND : Nondet ∈ es }} → Regex ↦es Tree
dmatch r = symbol >>= maybe
  (λ x → integralTree r <$> call (d r /d x))
  (if p ← ε? r then Pure (Sigma.fst p) else fail)
```

Here, *maybe f y* takes a *Maybe* value and applies f to the value in *just*, or returns y if it is *nothing*. Although the parser is easily seen to terminate in the intended semantics (since a character is

removed from the input string between each recursive call), a semantics where the call to *symbol* always returns *just* a character causes *dmatch* to diverge. That termination of *dmatch* is not a syntactical property, is reflected by the use of the *Rec* effect in its definition.

Adding the new effect *Parser* to our repertoire thus requires specifying its semantics. We gave the effects *Nondet* and *Rec* predicate transformer semantics in the form of a *PT* record. After introducing the *Parser* effect, the pre- and postcondition become more complicated: not only do they reference the 'pure' arguments and return values (here of type $r : \mathit{Regex}$ and $\mathit{Tree}\ r$ respectively), there is also the current state, containing a *String*, to keep track of. With these augmented predicates, the predicate transformer semantics for the *Parser* effect can be given as:

$$
\begin{aligned}
&\mathit{ptParser} : (c : \mathit{CParser}) \to (\mathit{RParser}\ c \to \mathit{String} \to \mathit{Set}) \to \mathit{String} \to \mathit{Set} \\
&\mathit{ptParser}\ \mathit{Symbol}\ P\ \mathit{Nil} \quad\ = P\ \mathit{nothing}\ \mathit{Nil} \\
&\mathit{ptParser}\ \mathit{Symbol}\ P\ (x :: \mathit{xs}) = P\ (\mathit{just}\ x)\ \mathit{xs}
\end{aligned}
$$

In this article, we want to demonstrate the modularity of predicate transformer semantics. To illustrate how the semantics mesh well with other forms of semantics, we do not use *ptParser* as semantics for *Parser* in the remainder. We give denotational semantics, in the form of an effect handler for *Parser* [23, 31]:

$$
\begin{aligned}
&\mathit{hParser} : \{\!\{\, \mathit{iND} : \mathit{Nondet} \in \mathit{es} \,\}\!\} \to (c : \mathit{CParser}) \to \mathit{String} \to \mathit{Free}\ \mathit{es}\ (\mathit{RParser}\ c \times \mathit{String}) \\
&\mathit{hParser}\ \mathit{Symbol}\ \mathit{Nil} \quad\ = \mathit{Pure}\ (\mathit{nothing}\ , \mathit{Nil}) \\
&\mathit{hParser}\ \mathit{Symbol}\ (x :: \mathit{xs}) = \mathit{Pure}\ (\mathit{just}\ x \quad , \mathit{xs})
\end{aligned}
$$

The function *handleRec* folds a given handler over a recursive definition, allowing us to handle the *Parser* effect in *dmatch*.

$$
\begin{aligned}
&\mathit{handleRec} : ((c : C) \to s \to \mathit{Free}\ \mathit{es}\ (R\ c \times s)) \to \\
&\quad a \overset{\mathit{mkSig}\ C\ R :: \mathit{es}}{\hookrightarrow} b \to (x : a \times s) \overset{\mathit{es}}{\hookrightarrow} b\ (\mathit{Pair.fst}\ x) \\
&\mathit{dmatch}' : \{\!\{\, \mathit{iND} : \mathit{Nondet} \in \mathit{es} \,\}\!\} \to (x : \mathit{Regex} \times \mathit{String}) \overset{\mathit{es}}{\hookrightarrow} \mathrm{Tree}\ (\mathit{Pair.fst}\ x) \\
&\mathit{dmatch}' = \mathit{handleRec}\ \mathit{hParser}\ (\mathit{dmatch})
\end{aligned}
$$

Note that *dmatch*′ has exactly the type of the previously defined *match*, conveniently allowing us to re-use the $[\![\_]\!]_{\mathrm{match}}$ semantics.

7 Proving total correctness

We finish the development process by proving that *dmatch* is correct. The first step in this proof is that *dmatch* always terminates. To express the termination of a recursive computation, we define the following predicate, terminates-in:

$$
\begin{aligned}
&\text{terminates-in} : (\mathit{pts} : \mathit{PTs}\ \mathit{es})\ (f : I \overset{\mathit{es}}{\hookrightarrow} O)\ (S : \mathit{Free}\ (\mathit{Rec}\ I\ O :: \mathit{es})\ a) \to \mathbb{N} \to \mathit{Set} \\
&\text{terminates-in}\ \mathit{pts}\ f\ (\mathit{Pure}\ x) \qquad\quad n \qquad\quad = \top \\
&\text{terminates-in}\ \mathit{pts}\ f\ (\mathit{Op} \in \mathrm{Head}\ c\ k)\ \mathit{Zero} \quad\ = \bot \\
&\text{terminates-in}\ \mathit{pts}\ f\ (\mathit{Op} \in \mathrm{Head}\ c\ k)\ (\mathit{Succ}\ n) = \text{terminates-in}\ \mathit{pts}\ f\ (f\ c \gg\!= k)\ n \\
&\text{terminates-in}\ \mathit{pts}\ f\ (\mathit{Op}\ (\in \mathrm{Tail}\ i)\ c\ k)\ n \quad\ = \\
&\quad \mathit{lookupPT}\ \mathit{pts}\ i\ c\ (\lambda\ x \to \text{terminates-in}\ \mathit{pts}\ f\ (k\ x)\ n)
\end{aligned}
$$

Given a program S that calls the recursive function $f : I \overset{es}{\nrightarrow} O$, we check whether the computation requires no more than a fixed number of steps to terminate.

Since *dmatch* always consumes a character before going in recursion, we can bound the number of recursive calls with the length of the input string. We formalize this argument in the lemma *dmatchTerminates*. Note that *dmatch′* is defined using the *hParser* handler, showing that we can mix denotational and predicate transformer semantics. The proof goes by induction on this string. Unfolding the recursive *call* gives *integralTree r* ⟨$⟩ *dmatch′* (*d r* /*d x* , *xs*), which we rewrite using the associativity monad law in a lemma called terminates-fmap.

```
dmatchTerminates : (r : Regex) (xs : String) →
  terminates-in (ptAll :: Nil) (dmatch′) (dmatch′ (r , xs)) (length xs)
dmatchTerminates r Nil with ε? r
dmatchTerminates r Nil | yes p  = tt
dmatchTerminates r Nil | no ¬p = tt
dmatchTerminates r (x :: xs) = terminates-fmap (length xs) (dmatch′ ((d r /d x) , xs))
  (dmatchTerminates (d r /d x) xs)
  where
  terminates-fmap : {f : I ⤳ O} {g : a → b} (n : ℕ) (S : Free (Rec I O :: es) a) →
    terminates-in pts f S n → terminates-in pts f (g ⟨$⟩ S) n
```

Apart from termination, correctness consists of soundness and completeness: the parse trees returned by *dmatch* should be satisfy the specification given by the original *Match* relation, and for any string that matches the regular expression, *dmatch* should return a parse tree. In the *ptAll* semantics, a nondeterministic program S is refined by T if and only if the output values of T are a subset of the output values of S; conversely S is refined by T in the *ptAny* semantics if and only if the output values of S are a subset of the output values of T. These properties allow us to express program correctness in terms of refinement.

We can show soundness of *dmatch* by proving it refines *match*. Transitivity of the refinement relation then allows us to conclude that it also satisfies the specification given by our original *Match* relation. The first step is to show that the derivative operator is correct, i.e. *d r* /*d x* matches those strings *xs* such that *r* matches *x* :: *xs*.

```
derivativeCorrect : ∀ r → Match (d r /d x) xs y → Match r (x :: xs) (integralTree r y)
```

The proof is straightforward by induction on the derivation of type *Match* (*d r* /*d x*) *xs y*.

Using the preceding lemmas, we can prove the partial correctness of *dmatch*.

```
dmatchSound : ∀ r xs → ⟦match (r , xs)⟧match ⊑ ⟦dmatch′ (r , xs)⟧match
```

Since we need to perform the case distinctions of *match* and of *dmatch*, the proof is longer than that of *matchSoundness*. Despite the length, most of it consists of this case distinction, then giving a simple argument for each case.

Although we successfully proved *dmatch* is sound with respect to the *Match* relation, it is not complete: the function *dmatch* never makes a non-deterministic choice. It will not return all possible parse trees that satisfy the *Match* relation, only the first tree that it encounters. We can, however, prove that *dmatch* will find a parse tree if it exists. To express that *dmatch* returns any result at all, we use a trivially true postcondition; by furthermore replacing the demonic

choice of the *ptAll* semantics with the angelic choice of *ptAny*, we require that *dmatch* must return a result:

```
dmatchComplete : ∀ r xs y → Match r xs y →
  ⟦dmatch′ (r , xs)⟧ptRec matchSpec :: ptAny :: Nil (λ _ → ⊤)
```

The proof is short, since *dmatch* can only *fail* when it encounters an empty string and a regular expression that does not match the empty string, which contradicts the assumption *Match r xs y*:

```
dmatchComplete r Nil y H with ε? r
... | yes p = tt
... | no ¬p = ¬p (_ , H)
dmatchComplete r (x :: xs) y H y′ H′ = tt
```

In the proofs of *dmatchSound* and *dmatchComplete*, we demonstrate the power of predicate transformer semantics for effects: by separating syntax and semantics, we can easily describe different aspects (soundness and completeness) of the one definition of *dmatch*. Since the soundness and completeness result we have proved imply partial correctness, and partial correctness and termination imply total correctness, we can conclude that *dmatch* is a totally correct parser for regular languages.

8 Discussion

Related work

The refinement calculus has traditionally been used in the verification of imperative programs [20]. In this paper, however, we show how many of the ideas from the refinement calculus can also be used in the verification of functional programs [29]. The Dijkstra monad, introduced in the language F⋆, also uses a predicate transformer semantics for verifying effectful programs by collecting the proof obligations for verification [26, 2, 17]. This paper demonstrates how similar verification efforts can be undertaken directly in an interactive theorem prover such as Agda. The separation of syntax and semantics in our approach allows for verification to be performed in several steps, such as we did for *dmatchTerminates*, *dmatchSound* and *dmatchComplete*, adding new effects as we need them.

Our running example of the regular expression parser is inspired by the development of a regular expression parser in by Harper [11]. More recently, Korkut, Trifunovski, and Licata [15] adapted the Functional Pearl to Agda. A direct translation of Harper's definitions is not possible: they are rejected by Agda's termination checker because they are not structurally recursive. Korkut, Trifunovski, and Licata show how the defunctionalization of Harper's matcher, written in continuation-passing style, is accepted by Agda's termination checker.

Formally verified parsers for a more general class of languages have been developed before: Danielsson [5], Ridge [24], and Firsov [7], among others, have previously shown how to verify parsers developed in a functional language. In these developments, semantics are defined specialized to the domain of parsing, while our semantics arise from combining a generic set of effect semantics. Furthermore, we allow our parsers to be written using general recursion directly, whereas most existing approaches deal with termination syntactically, either by incorporating delay and force operators in the grammar, or explicitly passing around a proof of termination

in the definition of the parser. The modularity of our setup allows us to separate partial and total correctness cleanly.

Open issues

This paper builds upon our previous results [29] by demonstrating their use in non-trivial development. In the process, we show how to combine predicate transformer semantics and reason about programs using a combination of effects.

Our approach relies on using coproducts to combine effect syntax. The interaction between different effects means applying handlers in a different order can result in different semantics. We assign predicate transformer semantics to a combination of effects all at once, specifying their interaction explicitly—but we would still like to explore how to handle effects one-by-one, allowing for greater flexibility when assigning semantics to effectful programs [31, 25].

Conclusions

In conclusion, we have illustrated the approach to developing verified software in a proof assistant using a predicate transformer semantics for effects for a non-trivial example. We believe this approach enables us to add new effects in a modular fashion, while still being able to re-use any existing proofs. Along the way, we introduced how to combine different effects and define different semantics for these effects, without impacting existing definitions. As a result, the verification effort—while conceptually more challenging at times—remains fairly modular.

Acknowledgements T. Baanen has received funding from the NWO under the Vidi program (project No. 016.Vidi.189.037, Lean Forward).

References

[1] Michael Abbott, Thorsten Altenkirch, and Neil Ghani. “Categories of Containers”. In: Proceedings of Foundations of Software Science and Computation Structures. 2003.

[2] Danel Ahman et al. “Dijkstra Monads for Free”. In: Proceedings of the 44th ACM SIGPLAN Symposium on Principles of Programming Languages. POPL 2017. Paris, France: Association for Computing Machinery, 2017, pp. 515–529. ISBN: 9781450346603. DOI: 10.1145/3009837.3009878.

[3] Andrej Bauer and Matija Pretnar. “Programming with algebraic effects and handlers”. In: Journal of Logical and Algebraic Methods in Programming 84.1 (2015). Special Issue: The 23rd Nordic Workshop on Programming Theory (NWPT 2011) Special Issue: Domains X, International workshop on Domain Theory and applications, Swansea, 5-7 September, 2011, pp. 108–123. ISSN: 2352-2208. DOI: https://doi.org/10.1016/j.jlamp.2014.02.001.

[4] Janusz A. Brzozowski. “Derivatives of Regular Expressions”. In: J. ACM 11.4 (Oct. 1964), pp. 481–494. ISSN: 0004-5411. DOI: 10.1145/321239.321249.

[5] Nils Anders Danielsson. “Total Parser Combinators”. In: SIGPLAN Not. 45.9 (Sept. 2010), pp. 285–296. ISSN: 0362-1340. DOI: 10.1145/1932681.1863585.

[6] Dominique Devriese and Frank Piessens. “On the Bright Side of Type Classes: Instance Arguments in Agda”. In: Proceedings of the 16th ACM SIGPLAN International Conference on Functional Programming. ICFP ’11. Tokyo, Japan: Association for Computing Machinery, 2011, pp. 143–155. ISBN: 9781450308656. DOI: 10.1145/2034773.2034796.

[7] Denis Firsov. “Certification of Context-Free Grammar Algorithms”. PhD thesis. Institute of Cybernetics at Tallinn University of Technology, 2016.

[8] Jeroen Fokker. “Functional Parsers”. In: Advanced Functional Programming, First International Spring School on Advanced Functional Programming Techniques-Tutorial Text. Berlin, Heidelberg: Springer-Verlag, 1995, pp. 1–23. ISBN: 3540594515.

[9] Richard A. Frost, Rahmatullah Hafiz, and Paul Callaghan. “Parser Combinators for Ambiguous Left-Recursive Grammars”. In: Proceedings of the 10th International Conference on Practical Aspects of Declarative Languages. PADL’08. San Francisco, CA, USA: Springer-Verlag, 2008, pp. 167–181. ISBN: 3540774416.

[10] Jeremy Gibbons and Ralf Hinze. “Just Do It: Simple Monadic Equational Reasoning”. In: SIGPLAN Not. 46.9 (Sept. 2011), pp. 2–14. ISSN: 0362-1340. DOI: 10.1145/2034574.2034777.

[11] Robert Harper. “Proof-directed debugging”. In: Journal of Functional Programming 9.4 (1999), pp. 463–469. DOI: 10.1017/S0956796899003378.

[12] Graham Hutton. “Higher-order functions for parsing”. In: Journal of Functional Programming 2.3 (1992), pp. 323–343. DOI: 10.1017/S0956796800000411.

[13] Graham Hutton and Diana Fulger. “Reasoning about effects: Seeing the wood through the trees”. 2008.

[14] Pieter W. M. Koopman and Marinus J. Plasmeijer. “Efficient Combinator Parsers”. In: Selected Papers from the 10th International Workshop on 10th International Workshop. IFL ’98. Berlin, Heidelberg: Springer-Verlag, 1998, pp. 120–136. ISBN: 3540662294.

[15] Joomy Korkut, Maksim Trifunovski, and Daniel R. Licata. “Intrinsic Verification of a Regular Expression Matcher”. preprint available at http://dlicata.web.wesleyan.edu/pubs/ktl16regexp/ktl16regexp.pdf. Jan. 2016.

[16] Daan Leijen and Erik Meijer. Parsec: Direct style monadic parser combinators for the real world. 2001.

[17] Kenji Maillard et al. “Dijkstra Monads for All”. In: Proc. ACM Program. Lang. 3.ICFP (July 2019). DOI: 10.1145/3341708.

[18] Conor McBride. “Turing-Completeness Totally Free”. In: Mathematics of Program Construction. Ed. by Ralf Hinze and Janis Voigtländer. Cham: Springer International Publishing, 2015, pp. 257–275. ISBN: 978-3-319-19797-5.

[19] Matthew Might, David Darais, and Daniel Spiewak. “Parsing with Derivatives: A Functional Pearl”. In: Proceedings of the 16th ACM SIGPLAN International Conference on Functional Programming. ICFP ’11. Tokyo, Japan: Association for Computing Machinery, 2011, pp. 189–195. ISBN: 9781450308656. DOI: 10.1145/2034773.2034801.

[20] Caroll Morgan. Programming from Specifications. 2nd ed. Prentice Hall, 1998.

[21] Ulf Norell. “Towards a practical programming language based on dependent type theory”. PhD thesis. Chalmers University of Technology, 2007.

[22] Gordon Plotkin and John Power. “Algebraic Operations and Generic Effects”. In: Applied Categorical Structures 11.1 (Feb. 2003), pp. 69–94. ISSN: 1572-9095. DOI: 10.1023/A:1023064908962.

[23] Gordon Plotkin and Matija Pretnar. “Handlers of Algebraic Effects”. In: Proceedings of the 18th European Symposium on Programming Languages and Systems: Held as Part of the Joint European Conferences on Theory and Practice of Software, ETAPS 2009. ESOP ’09. York, UK: Springer-Verlag, 2009, pp. 80–94. ISBN: 9783642005893. DOI: 10.1007/978-3-642-00590-9_7.

[24] Tom Ridge. “Simple, Functional, Sound and Complete Parsing for All Context-Free Grammars”. In: Certified Programs and Proofs. Ed. by Jean-Pierre Jouannaud and Zhong Shao. Berlin, Heidelberg: Springer Berlin Heidelberg, 2011, pp. 103–118. ISBN: 978-3-642-25379-9.

[25] Tom Schrijvers et al. “Monad Transformers and Modular Algebraic Effects: What Binds Them Together”. In: Proceedings of the 12th ACM SIGPLAN International Symposium on Haskell. Haskell 2019. Berlin, Germany: Association for Computing Machinery, 2019, pp. 98–113. ISBN: 9781450368131. DOI: 10.1145/3331545.3342595.

[26] Nikhil Swamy et al. “Verifying Higher-order Programs with the Dijkstra Monad”. In: Proceedings of the 34th ACM SIGPLAN Conference on Programming Language Design and Implementation. PLDI ’13. Seattle, Washington, USA: ACM, 2013, pp. 387–398. ISBN: 978-1-4503-2014-6. DOI: 10.1145/2491956.2491978.

[27] S. Doaitse Swierstra and Luc Duponcheel. “Deterministic, Error-Correcting Combinator Parsers”. In: Advanced Functional Programming. Springer-Verlag, 1996, pp. 184–207.

[28] Wouter Swierstra. “Data types à la carte”. In: Journal of Functional Programming 18.4 (2008), pp. 423–436. DOI: 10.1017/S0956796808006758.

[29] Wouter Swierstra and Tim Baanen. “A predicate transformer semantics for effects (Functional Pearl)”. In: Proceedings of the 24th ACM SIGPLAN International Conference on Functional Programming. ICFP ’19. 2019. DOI: 10.1145/3341707.

[30] Philip Wadler. “How to Replace Failure by a List of Successes”. In: Proc. Of a Conference on Functional Programming Languages and Computer Architecture. Nancy, France: Springer-Verlag New York, Inc., 1985, pp. 113–128.

[31] Nicolas Wu, Tom Schrijvers, and Ralf Hinze. “Effect Handlers in Scope”. In: Proceedings of the 2014 ACM SIGPLAN Symposium on Haskell. Haskell ’14. Gothenburg, Sweden: ACM, 2014, pp. 1–12. ISBN: 978-1-4503-3041-1. DOI: 10.1145/2633357.2633358.